

Chapter 10

Control Break Logic

Objectives

Upon completion of this chapter you will be able to:

- Describe the write, roll, zero, and save steps used in control break processing,
- Design and code a program using single level control break processing,
- Design and code a program using multiple level control break processing, and
- Design and code a program using control break processing to produce a summary report.

Introduction

In each of our programs thus far any counts and/or totals were shown at end of job only. In this chapter we will learn to produce **subtotals** as well. Subtotals can be done at any number of levels, and are the result of **control break processing**. In this chapter we will introduce single level and multiple level control break processing. No new instructions are introduced in this chapter.

Throughout this chapter we will make use of the course offerings file, `OFFER.DAT`. Our first program, `OFFER10A.MLC`, lists selected fields from each record in that file. Page break logic is included. There is a limit of ten detail lines per page. The output is shown below, and the complete program listing is found on the next page. There is nothing new here: the program is given so as to provide a starting point for our discussion.

COURSE OFFERINGS				Page	1
Sem	Course	Section	Teacher		
---	-----	-----	-----		
W92	EG102	1	732		
W92	MA107	1	218		
W92	PE151	1	574		
F92	AC101	1	218		
F92	BU101	1	218		
F92	EG101	1	732		
F92	EG101	2	732		
F92	MA101	1	626		
F92	MA101	2	626		
F92	PE151	1	574		

COURSE OFFERINGS				Page	2
Sem	Course	Section	Teacher		
---	-----	-----	-----		
W93	EG102	1	854		
W93	MA101	1	626		
W93	MA107	1	626		
W93	PE151	1	574		

```

                PRINT NOGEN
*****
*      FILENAME:  OFFER10A.MLC      *
*      AUTHOR   :  Bill Qualls     *
*      SYSTEM   :  PC/370 R4.2     *
*      REMARKS  :  Produce list of course offerings for all *
*                  semesters. Includes page break logic.   *
*****
                START 0
                REGS
BEGIN          BEGIN
                WTO    'OFFER10A ... Begin execution'
                BAL    R10,SETUP
MAIN          EQU    *
                CLI    EOFSW,C'Y'
                BE     EOJ
                BAL    R10,PROCESS
                B      MAIN
EOJ           EQU    *
                BAL    R10,WRAPUP
                WTO    'OFFER10A ... Normal end of program'
                RETURN
*****
*      SETUP - Those things which happen one time only,   *
*              before any records are processed.          *
*****
SETUP        EQU    *
                ST     R10,SVSETUP
                OI     OFFER+10,X'08'      PC/370 ONLY - Convert all
*                                                    input from ASCII to EBCDIC
                OI     REPORT+10,X'08'    PC/370 ONLY - Convert all
*                                                    output from EBCDIC to ASCII
                OPEN  OFFER
                OPEN  REPORT
                BAL    R10,READ
                L      R10,SVSETUP
                BR     R10
*****
*      HDGS - Print headings.                             *
*****
HDGS        EQU    *
                ST     R10,SVHDGS
                AP     PGS,=P'1'          Add 1 to page count
                MVC    HDPGS,=X'40202120' Edit pattern for page count
                ED     HDPGS,PGS         Move page count to heading
                PUT    REPORT,FORMFEED   PC/370 ONLY
                PUT    REPORT,HD1
                PUT    REPORT,HD2
                PUT    REPORT,HD3
                PUT    REPORT,HD4
                ZAP    LNS,=P'0'         Reset line count to zero
                L      R10,SVHDGS
                BR     R10
*****
*      PROCESS - Those things which happen once per record. *
*****
PROCESS     EQU    *
                ST     R10,SVPROC
                BAL    R10,CHKLNS
                BAL    R10,FORMAT

```

(continued)

```

        BAL    R10,WRITE
        BAL    R10,READ
        L      R10,SVPROC
        BR     R10
*****
*          READ - Read a record.          *
*****
READ      EQU    *
          ST     R10,SVREAD
          GET    OFFER,IREC           Read a single offer record
          B      READX
ATEND     EQU    *
          MVI    EOFSW,C'Y'
READX     EQU    *
          L      R10,SVREAD
          BR     R10
*****
*          CHKLNS - Check lines printed. Full page?  *
*****
CHKLNS    EQU    *
          ST     R10,SVCHKLNS
          CP     LNS,MAXLNS
          BL     CHKLNSX
          BAL    R10,HDGS
CHKLNSX   EQU    *
          L      R10,SVCHKLNS
          BR     R10
*****
*          FORMAT - Format a single detail line.      *
*****
FORMAT    EQU    *
          ST     R10,SVFORM
          MVC    OREC(40),BLANKS
          MVC    OSEM,ISEM           Semester
          MVC    OCID,ICID           Course ID
          MVC    OSECT,ISECT         Section number
          MVC    OTID,ITID           Teacher ID
          MVC    OCRLF,WCRLF         PC/370 Only
          L      R10,SVFORM
          BR     R10
*****
*          WRITE - Write a single detail line.        *
*****
WRITE     EQU    *
          ST     R10,SVWRITE
          PUT    REPORT,OREC         Write report line
          AP     LNS,=P'1'
          L      R10,SVWRITE
          BR     R10
*****
*          WRAPUP - Those things which happen one time only,
*          after all records have been processed.    *
*****
WRAPUP    EQU    *
          ST     R10,SVWRAP
          CLOSE OFFER
          CLOSE REPORT
          WTO    'OFFER10A ... Course list on REPORT.TXT'
          L      R10,SVWRAP
          BR     R10

```

(continued)

```

*****
*           Literals, if any, will go here           *
*****
                LTORG
*****
*           File definitions                         *
*****
OFFER    DCB    LRECL=18,RECFM=F,MACRF=G,EODAD=ATEND,
                DDNAME='OFFER.DAT'
REPORT   DCB    LRECL=42,RECFM=F,MACRF=P,
                DDNAME='REPORT.TXT'
*****
*           RETURN ADDRESSES                       *
*****
SVSETUP  DC     F'0'                SETUP
SVHDGS   DC     F'0'                HDGS
SVPROC   DC     F'0'                PROCESS
SVREAD   DC     F'0'                READ
SVFORM   DC     F'0'                FORMAT
SVWRITE  DC     F'0'                WRITE
SVWRAP   DC     F'0'                WRAPUP
SVCHKLNS DC     F'0'                CHKLNS
*****
*           Miscellaneous field definitions         *
*****
WCRLF    DC     X'0D25'             PC/370 ONLY - EBCDIC CR/LF
EOFSW    DC     CL1'N'              End of file? (Y/N)
PGS       DC     PL2'0'             Nbr of pages printed.
LNS       DC     PL2'10'           Lines printed on this page.
MAXLNS   DC     PL2'10'           Max nbr lines per page.
*
BLANKS   DC     CL40' '            My line counts exclude hdgs.
*****
*           Input record definition                *
*****
IREC     DS     0CL18             1-18   Offer record
ISEM     DS     CL3               1- 3   Semester
ICID     DS     0CL5             4- 8   Course ID
IDEPT    DS     CL2              4- 5   Department
         DS     CL3              6- 8   Course number
ISECT    DS     CL1              9- 9   Section number
ITID     DS     CL3             10-12  Teacher ID
IROOM    DS     CL4             13-16  Room number
IOCRFLF  DS     CL2             17-18  PC/370 only - CR/LF
*****
*           Output (line) definition              *
*****
OREC     DS     0CL42             1-42
         DC     CL3' '            1- 3
OSEM     DS     CL3              4- 6   Semester
         DC     CL4' '            7-10
OCID     DS     CL5             11-15  Course ID
         DC     CL6' '            16-21
OSECT    DS     CL1             22-22  Section number
         DC     CL8' '            23-30
OTID     DS     CL3             31-33  Teacher ID
         DC     CL7' '            34-40
OCRLF    DS     CL2             41-42  PC/370 only - CR/LF

```

(continued)

```

*****
*           Headings definitions           *
*****
*
*  -----1-----2-----3-----4
*           COURSE OFFERINGS           PageBZZ9
*
*  Sem      Course      Section      Teachers
*  ---      -
*  XXX      XXXXX       X            XXX
*  XXX      XXXXX       X            XXX
*  XXX      XXXXX       X            XXX
*
FORMFEED DS      0CL42                PC/370 only
*         DC      X'0C'                EBCDIC formfeed
*         DC      CL39' '
*         DC      40C' '                For testing...
*         DC      X'0D25'                EBCDIC CR/LF
HD1      DS      0CL42
*         DC      CL36'                COURSE OFFERINGS           Page'
HDPGS    DC      CL4'BZZ9'
*         DC      XL2'0D25'
HD2      DS      0CL42
*         DC      CL40' '
*         DC      XL2'0D25'
HD3      DS      0CL42
*         DC      CL40' Sem Course Section Teacher '
*         DC      XL2'0D25'
HD4      DS      0CL42
*         DC      CL40' --- -----
*         DC      XL2'0D25'
*         END      BEGIN

```

Single Level Control Break - Programming Problem

Notice that the above report is in semester sequence. *Let's modify the report so that there is only one semester per page.* Since there will be only one semester per page, it would be redundant to include the semester in the detail lines, so let's drop that column, and show the semester in the headings instead. Also, let's show a count of the number of sections in a semester at the end of each semester. Our new report will appear as follows:

```

-----
COURSE OFFERINGS           Page 1
Semester W92
Course      Section      Teacher
-----
EG102       1            732
MA107       1            218
PE151       1            574
*** Sem W92      3 sections
-----

```

<---<< Semester
in heading

<---<< Count by
semester

(continued)

COURSE OFFERINGS Semester F92	Page 2	<--<< One semester per page																											
<table style="width: 100%; border-collapse: collapse;"> <thead> <tr> <th style="text-align: left;">Course</th> <th style="text-align: left;">Section</th> <th style="text-align: left;">Teacher</th> </tr> <tr> <th style="text-align: left;">-----</th> <th style="text-align: left;">-----</th> <th style="text-align: left;">-----</th> </tr> </thead> <tbody> <tr><td>AC101</td><td>1</td><td>218</td></tr> <tr><td>BU101</td><td>1</td><td>218</td></tr> <tr><td>EG101</td><td>1</td><td>732</td></tr> <tr><td>EG101</td><td>2</td><td>732</td></tr> <tr><td>MA101</td><td>1</td><td>626</td></tr> <tr><td>MA101</td><td>2</td><td>626</td></tr> <tr><td>PE151</td><td>1</td><td>574</td></tr> </tbody> </table>	Course	Section	Teacher	-----	-----	-----	AC101	1	218	BU101	1	218	EG101	1	732	EG101	2	732	MA101	1	626	MA101	2	626	PE151	1	574		
Course	Section	Teacher																											
-----	-----	-----																											
AC101	1	218																											
BU101	1	218																											
EG101	1	732																											
EG101	2	732																											
MA101	1	626																											
MA101	2	626																											
PE151	1	574																											
*** Sem F92	7 sections																												

COURSE OFFERINGS Semester W93	Page 3																		
<table style="width: 100%; border-collapse: collapse;"> <thead> <tr> <th style="text-align: left;">Course</th> <th style="text-align: left;">Section</th> <th style="text-align: left;">Teacher</th> </tr> <tr> <th style="text-align: left;">-----</th> <th style="text-align: left;">-----</th> <th style="text-align: left;">-----</th> </tr> </thead> <tbody> <tr><td>EG102</td><td>1</td><td>854</td></tr> <tr><td>MA101</td><td>1</td><td>626</td></tr> <tr><td>MA107</td><td>1</td><td>626</td></tr> <tr><td>PE151</td><td>1</td><td>574</td></tr> </tbody> </table>	Course	Section	Teacher	-----	-----	-----	EG102	1	854	MA101	1	626	MA107	1	626	PE151	1	574	
Course	Section	Teacher																	
-----	-----	-----																	
EG102	1	854																	
MA101	1	626																	
MA107	1	626																	
PE151	1	574																	
*** Sem W93	4 sections																		

Single Level Control Break - Logic

The changes necessary to produce this report is as follows:

1. Check for a change in semester. In order to see if the semester in one record differs from the semester in the previous record, we will need a variable wherein we can "hold" the semester from record to record. This will be defined as:

```
HOLDSEM DC CL3 ' ' Hold semester
```

We said the semester would now be a part of the headings:

```
HD2 DS 0CL42
DC CL21 ' Semester '
HDSEM DS CL3
DC CL16' '
DC XL2 '0D25'
```

...so prior to printing the headings, we will move HOLDSEM to HDSEM:

```
HDGS EQU *
ST R10,SVHDGS
AP PGS,=P'1' Add 1 to page count
MVC HDPGS,=X'40202120' Edit pattern for page count
ED HDPGS,PGS Move page count to heading
PUT REPORT,FORMFEED PC/370 ONLY
MVC HDSEM,HOLDSEM
PUT REPORT,HD1
```

Of course, we want the semester from the first record to be recognized as such, otherwise we will try to print totals after having read the first record only. So, we initialize `HOLDSEM` within the `SETUP` routine immediately after the priming read:

```
BAL R10, READ
MVC HOLDSEM, ISEM      Control break
```

As each record is processed, we will check for a change in semester. Since this is a check done for each record, it will, of course, fall within the `PROCESS` routine (as opposed to `SETUP` or `WRAPUP`.) We will invoke a separate routine, `CHKSEM`, to see if the semester has changed. We do so only to keep the `PROCESS` routine simple and uncluttered:

```
PROCESS EQU *
ST R10, SVPROC
BAL R10, CHKSEM      See if new semester
```

...where `CHKSEM` is defined as:

```
*****
*          CHKSEM - Check for change in semester          *
*                    (control break)                    *
*****
CHKSEM EQU *
ST R10, SVCHKSEM
CLC HOLDSEM, ISEM      Compare w/ current
BE CHKSEM, X           Same semester, get out
BAL R10, ENDSEM        Process semester break
MVC HOLDSEM, ISEM      Update control break field
CHKSEM EQU *
L R10, SVCHKSEM
BR R10
```

Again, this could have been done within `PROCESS` as follows:

```
PROCESS EQU *
ST R10, SVPROC
CLC HOLDSEM, ISEM      Compare w/ current
BE PROCESS2, X         Same semester, get out
BAL R10, ENDSEM        Process semester break
MVC HOLDSEM, ISEM      Update control break field
PROCESS2 EQU *
```

(continued)

Note that having ended one semester and started another, we move the current semester, `ISEM`, to the control break field, `HOLDSEM`.

2. If the semester has changed, then end that semester and start another. We see from `CHKSEM` that if the semester has changed, that is, if `HOLDSEM` does not equal `ISEM`, then we invoke the control break routine, `ENDSEM`. This new routine will (a) print the number of sections for this semester, (b) reset the sections counter to zero, and (c) force the next semester to a new page:

```

ENDSEM  EQU  *
        ST   R10,SVENDSEM
        MVC  OREC(40),BLANKS      This area used several ways
        BAL  R10,WRITE           Skip a line
        MVC  OREC+6(25),=C'*** Sem XXX BZZ9 sections'
        MVC  OREC+14(3),HOLDSEM
        MVC  OREC+18(4),=X'40202120'
        ED   OREC+18(4),#SEM
        BAL  R10,WRITE
        MVC  OREC(40),BLANKS
        ZAP  #SEM,=P'0'          Reset counter
        ZAP  LNS,MAXLNS         Force next sem. to new page
        L    R10,SVENDSEM
        BR   R10
    
```

- 2(a) Print the number of sections for this semester. Note how the count is formatted. First, the line is cleared. (This is the same area as was used for the details, but we can reuse it for showing the counts if we like.) Of course, when we move `BLANKS` to `OREC`, we do so for a length of 40 only so as not to destroy the `CR/LF`. Otherwise, we would need to remember to move `X'0D25'` to `OREC+40(2)`. We then move a description of the count. Notice how a long label was used, with `XXX` indicating where the semester will go, and `BZZ9` indicating where the count will go. It is by no means necessary to do it this way. I find this technique to be quite useful in that it is somewhat self-documenting. This is particularly important if we are going to use explicit displacement and length as shown above. Nevertheless, the following is functionally equivalent and probably more commonly used:

```

MVC  OREC+6(7),=C'*** Sem'
MVC  OREC+14(3),HOLDSEM
MVC  OREC+18(4),=X'40202120'
ED   OREC+18(4),#SEM
MVC  OREC+23(8),=C'sections'
    
```

Having used the detail line in this way, we must remember to clean up after ourselves; that is, once again move `BLANKS` to `OREC(40)`.

You might feel uneasy with reusing the detail line as we have done here. It is very common to define a separate record for this purpose:

```

MVC  TOTSEM,HOLDSEM
MVC  TOTCOUNT,=X'40202120'
ED   TOTCOUNT,#SEM
PUT  REPORT,TOTREC
AP   LNS,=P'1'
    
```

where...

```
TOTREC  DS    0CL42
        DC    CL6'  '
        DC    CL8'*** SEM '
TOTSEM  DC    CL3'  '
        DC    CL1'  '
TOTCOUNT DC  CL4'  '
        DC    CL18' sections'
        DC    X'0D25' CR/LF
```

Note that we use `PUT REPORT, TOTREC` and increment the line counter here rather than `BAL R10, WRITE` because that (`WRITE`) routine puts `OREC` (only) to `REPORT`.

- 2(b) Reset the sections counter to zero. We move zero to the counter for number of sections within this semester, `#SEM`. Otherwise, the sections count for semester `W92` will be included in the sections count for semester `F93`, and the sections count for semesters `W92` and `F93` will be included in the sections count for semester `W93`.
- 2(c) Force the next semester to a new page. Forcing each new semester to a new page is simple: just set the line counter to the maximum allowed. Recall that before printing any detail lines, we always invoke the `CHKLNS` routine to see if a page is full and, if so, print the headings. By setting the line counter to the maximum allowed, the page will be considered full and headings will be printed, with the new semester, prior to printing the next detail line.
3. As each record is processed, add one to the number of sections for that semester. This is done for each record and, as such, is included in the `PROCESS` routine. It is done after the check for a new semester, otherwise the wrong semester may get "credit" for this record.

```
PROCESS EQU  *
        ST   R10, SVPROC
        BAL  R10, CHKSEM          See if new semester
        AP   #SEM, =P'1'        Count sections by semester
```

4. After all records have been processed, print one last set of totals. Within the `WRAPUP` routine we invoke the control break routine one last time: not because a new semester has begun, but because the last semester has ended. If we fail to do so, then we will not get a count of sections for the last semester.

```
WRAPUP EQU  *
        ST   R10, SVWRAP
        BAL  R10, ENDSEM        Final control break process
        CLOSE OFFER
        CLOSE REPORT
```

This can be done before or after closing the `OFFER` file, but *must* be done *before* closing the `REPORT` file.

Single Level Control Break - Program Solution

```

PRINT NOGEN
*****
*      FILENAME:  OFFER10B.MLC      *
*      AUTHOR   :  Bill Qualls     *
*      SYSTEM   :  PC/370 R4.2     *
*      REMARKS  :  Produce list of course offerings for all *
*                  semesters. Includes page break logic.  *
*      Single level control break example. *
*****
      START 0
      REGS
BEGIN  BEGIN
      WTO   'OFFER10B ... Begin execution'
      BAL   R10,SETUP
MAIN   EQU   *
      CLI   EOFSW,C'Y'
      BE    EOJ
      BAL   R10,PROCESS
      B     MAIN
EOJ    EQU   *
      BAL   R10,WRAPUP
      WTO   'OFFER10B ... Normal end of program'
      RETURN
*****
*      SETUP - Those things which happen one time only, *
*                  before any records are processed.   *
*****
SETUP  EQU   *
      ST    R10,SVSETUP
      OI    OFFER+10,X'08'      PC/370 ONLY - Convert all
*                                     input from ASCII to EBCDIC
      OI    REPORT+10,X'08'    PC/370 ONLY - Convert all
*                                     output from EBCDIC to ASCII
      OPEN  OFFER
      OPEN  REPORT
      BAL   R10,READ
      MVC   HOLDSEM,ISEM      Control break
      L     R10,SVSETUP
      BR    R10
*****
*      HDGS - Print headings. *
*****
HDGS   EQU   *
      ST    R10,SVHDGS
      AP    PGS,=P'1'          Add 1 to page count
      MVC   HDPGS,=X'40202120' Edit pattern for page count
      ED    HDPGS,PGS          Move page count to heading
      PUT   REPORT,FORMFEED    PC/370 ONLY
      MVC   HDSEM,HOLDSEM
      PUT   REPORT,HD1
      PUT   REPORT,HD2
      PUT   REPORT,HD3
      PUT   REPORT,HD4
      PUT   REPORT,HD5
      ZAP   LNS,=P'0'          Reset line count to zero
      L     R10,SVHDGS
      BR    R10

```

(continued)

```

*****
*      PROCESS - Those things which happen once per record.  *
*****
PROCESS EQU *
      ST  R10,SVPROC
      BAL R10,CHKSEM          See if new semester
      AP  #SEM,=P'1'         Count sections by semester
      BAL R10,CHKLNS
      BAL R10,FORMAT
      BAL R10,WRITE
      BAL R10,READ
      L   R10,SVPROC
      BR  R10
*****
*      READ - Read a record.  *
*****
READ EQU *
      ST  R10,SVREAD
      GET OFFER,IREC          Read a single offer record
      B   READX
ATEND EQU *
      MVI EOFSW,C'Y'
READX EQU *
      L   R10,SVREAD
      BR  R10
*****
*      CHKSEM - Check for change in semester  *
*      (control break)  *
*****
CHKSEM EQU *
      ST  R10,SVCHKSEM
      CLC HOLDSEM,ISEM        Compare w/ current
      BE  CHKSEM             Same semester, get out
      BAL R10,ENDSEM         Process semester break
      MVC HOLDSEM,ISEM       Update control break field
CHKSEM EQU *
      L   R10,SVCHKSEM
      BR  R10
*****
*      ENDSEM - End semester  *
*      (Process control break)  *
*      Show count of sections for this semester.  *
*      Force next semester to another page.  *
*****
ENDSEM EQU *
      ST  R10,SVENDSEM
      MVC OREC(40),BLANKS     This area used several ways
      BAL R10,WRITE           Skip a line
      MVC OREC+6(25),=C'*** Sem XXX BZZ9 sections'
      MVC OREC+14(3),HOLDSEM
      MVC OREC+18(4),=X'40202120'
      ED  OREC+18(4),#SEM
      BAL R10,WRITE
      MVC OREC(40),BLANKS
      ZAP #SEM,=P'0'          Reset counter
      ZAP LNS,MAXLNS          Force next sem. to new page
      L   R10,SVENDSEM
      BR  R10

```

(continued)

```

*****
*          CHKLNS - Check lines printed. Full page?          *
*****
CHKLNS   EQU      *
         ST       R10,SVCHKLNS
         CP       LNS,MAXLNS
         BL       CHKLNSX
         BAL      R10,HDGS
CHKLNSX  EQU      *
         L        R10,SVCHKLNS
         BR       R10
*****
*          FORMAT - Format a single detail line.             *
*****
FORMAT   EQU      *
         ST       R10,SVFORM
         MVC      OREC(40),BLANKS      Removed MVC OSEM, ISEM
         MVC      OCID,ICID           Course ID
         MVC      OSECT,ISECT        Section number
         MVC      OTID,ITID          Teacher ID
         MVC      OCRLF,WCRRLF       PC/370 Only
         L        R10,SVFORM
         BR       R10
*****
*          WRITE - Write a single detail line.              *
*****
WRITE    EQU      *
         ST       R10,SVWRITE
         PUT      REPORT,OREC         Write report line
         AP       LNS,=P'1'
         L        R10,SVWRITE
         BR       R10
*****
*          WRAPUP - Those things which happen one time only, *
*          after all records have been processed.          *
*****
WRAPUP   EQU      *
         ST       R10,SVWRAP
         BAL      R10,ENDSEM           Final control break process
         CLOSE   OFFER
         CLOSE   REPORT
         WTO      'OFFER10B ... Course list on REPORT.TXT'
         L        R10,SVWRAP
         BR       R10
*****
*          Literals, if any, will go here                  *
*****
         LTORG
*****
*          File definitions                                  *
*****
OFFER    DCB      LRECL=18,RECFM=F,MACRF=G,EODAD=ATEND,
         DDNAME='OFFER.DAT'
REPORT   DCB      LRECL=42,RECFM=F,MACRF=P,
         DDNAME='REPORT.TXT'
*****
*          RETURN ADDRESSES                                *
*****
SVSETUP  DC       F'0'                SETUP
SVHDGS   DC       F'0'                HDGS

```

(continued)

```

SVPROC  DC  F'0'          PROCESS
SVREAD  DC  F'0'          READ
SVFORM  DC  F'0'          FORMAT
SVWRITE DC  F'0'          WRITE
SVWRAP  DC  F'0'          WRAPUP
SVCHKLNS DC F'0'          CHKLNS
SVCHKSEM DC  F'0'          CHKSEM
SVENDSEM DC  F'0'          ENDSEM
*****
*           Miscellaneous field definitions           *
*****
WCRLF   DC  X'0D25'       PC/370 ONLY - EBCDIC CR/LF
EOFSW   DC  CL1'N'        End of file? (Y/N)
PGS     DC  PL2'0'        Nbr of pages printed.
LNS     DC  PL2'10'       Lines printed on this page.
MAXLNS  DC  PL2'10'       Max nbr lines per page.
*                               My line counts exclude hdgs.
BLANKS  DC  CL40' '
HOLDSEM DC  CL3' '       Hold semester
#SEM    DC  PL2'0'       Sections in a semester
*****
*           Input record definition                 *
*****
IREC    DS  0CL18         1-18  Offer record
ISEM    DS  CL3           1- 3  Semester
ICID    DS  0CL5         4- 8  Course ID
IDEPT   DS  CL2          4- 5  Department
ISect   DS  CL3          6- 8  Course number
ISECT   DS  CL1          9- 9  Section number
ITID    DS  CL3         10-12  Teacher ID
IROOM   DS  CL4         13-16  Room number
IOCRLF  DS  CL2         17-18  PC/370 only - CR/LF
*****
*           Output (line) definition                 *
*****
OREC    DS  0CL42         1-42
DC  CL10' '       1-10
OCID    DS  CL5          11-15  Course ID
        DC  CL6' '       16-21
Osect   DS  CL1          22-22  Section number
        DC  CL8' '       23-30
OTID    DS  CL3          31-33  Teacher ID
        DC  CL7' '       34-40
OCRLF   DS  CL2         41-42  PC/370 only - CR/LF
*****
*           Headings definitions                     *
*****
*
*  ----+----1----+----2----+----3----+----4
*                COURSE OFFERINGS          PageBZZ9
*
* Semester XXX
*
*           Course      Section      Teachers
*           -----      -
*           XXXXX        X           XXX
*           XXXXX        X           XXX
*           XXXXX        X           XXX
*
* Sem XXX BZZ9 sections
*

```

(continued)

```

FORMFEED DS      0CL42                PC/370 only
*         DC      X'0C'                EBCDIC formfeed
*         DC      CL39' '
         DC      40C' '                For testing...
         DC      X'0D25'                EBCDIC CR/LF
HD1       DS      0CL42
         DC      CL36'                COURSE OFFERINGS      Page '
HDFGS     DC      CL4'BZZ9'
         DC      XL2'0D25'
HD2       DS      0CL42
         DC      CL21'                Semester '
HDSEM     DS      CL3
         DC      CL16' '
         DC      XL2'0D25'
HD3       DS      0CL42
         DC      CL40' '
         DC      XL2'0D25'
HD4       DS      0CL42
         DC      CL40'                Course   Section  Teacher  '
         DC      XL2'0D25'
HD5       DS      0CL42
         DC      CL40'                -----  -----  -----  '
         DC      XL2'0D25'
END       BEGIN

```

Multiple Level Control Break - Programming Problem

We now introduce *multiple level control break* processing. Let's assume that, in addition to showing a count of sections by semester, we would also like to show a count of sections *by department within semester*; for example, after listing all sections offered by the Math department, we will show a count of those sections. We will, however, show this count only if the count is greater than 1, so as not to clutter our report. Furthermore, we will always double space (skip a line) between departments. Our new report will appear as shown on the next page.

Multiple Level Control Break - Logic

The logic necessary to produce this report is similar to that used in single level control break processing and is as follows:

1. Check for a change in department. Just as we checked for a change in semester in the previous program, we need to check for a change in department in this program. In order to do so, we will need a variable wherein we can "hold" the department from record to record. This will be defined as:

```

HOLDDEPT DC      CL2' '                Hold department

```

Of course, we want the semester *and* department from the first record to be recognized as such, so we initialize HOLDSEM *and* HOLDDEPT within the SETUP routine after the priming read:

Output from OFFER10C.MLC - Multiple Level Control Break Example

COURSE OFFERINGS			Page
Semester W92			1
Course	Section	Teacher	
EG102	1	732	
MA107	1	218	
PE151	1	574	
*** Sem W92	3 sections		

COURSE OFFERINGS			Page
Semester F92			2
Course	Section	Teacher	
AC101	1	218	
BU101	1	218	
EG101	1	732	
EG101	2	732	

**** Dept EG 2 sections**

MA101	1	626
-------	---	-----

COURSE OFFERINGS			Page
Semester F92			3
Course	Section	Teacher	
MA101	2	626	

*Note that semester F92
"spans" two pages.*

**** Dept MA 2 sections**

PE151	1	574
-------	---	-----

*** Sem F92 7 sections

COURSE OFFERINGS			Page
Semester W93			4
Course	Section	Teacher	
EG102	1	854	
MA101	1	626	
MA107	1	626	

**** Dept MA 2 sections**

PE151	1	574
-------	---	-----

*** Sem W93 4 sections

BAL	R10, READ	
MVC	HOLDSEM, ISEM	Major control break
MVC	HOLDDEPT, IDEPT	Minor control break

As each record is processed, we will check for a change in semester *or* department. Since this is a check done for each record, it will fall within the `PROCESS` routine. We will invoke `CHKSEM` to see if the semester has changed, and invoke `CHKDEPT` to see if the department has changed.

It is important to note that a change in semester implies a change in department. For example, if the last department listed in semester `W92` was `PE`, and the first department listed in semester `F93` was also `PE`, then these would appear on two separate pages and have two separate counts. *So we check the semester before checking the department:*

PROCESS	EQU	*	
	ST	R10, SVPROC	
	BAL	R10, CHKSEM	See if new semester
	BAL	R10, CHKDEPT	See if new department

where `CHKSEM` and `CHKDEPT` are...

```
*****
*      CHKSEM - Check for change in semester          *
*      (major control break)                          *
*****
CHKSEM  EQU      *
        ST       R10, SVCHKSEM
        CLC      HOLDSEM, ISEM          Compare w/ current
        BE       CHKSEM                Same semester, get out
        BAL      R10, ENDSEM           Process semester break
        MVC      HOLDSEM, ISEM          Update major break field
        MVC      HOLDDEPT, IDEPT       Update minor break field
CHKSEM  EQU      *
        L        R10, SVCHKSEM
        BR       R10
*****
*      CHKDEPT - Check for change in department      *
*      (minor control break)                          *
*****
CHKDEPT EQU      *
        ST       R10, SVCHKDEP
        CLC      HOLDDEPT, IDEPT       Compare w/ current
        BE       CHKDEPTX             Same department, get out
        BAL      R10, ENDDEPT         Process department break
        MVC      HOLDDEPT, IDEPT       Update control break field
CHKDEPT EQU      *
        L        R10, SVCHKDEP
        BR       R10
```

In multiple level control break processing, always evaluate the control break fields from major to minor!

2. If the semester has changed, then end that semester and start another. The processing here is the same as in the previous program except *a change in semester implies a change in department*. So the first thing we do within ENDSEM is end the previous department and begin the next department:

```

ENDSEM  EQU  *
        ST  R10,SVENDSEM
        BAL R10,ENDDEPT      Change in semester implies
*                               change in department as well.
        MVC OREC(40),BLANKS  This area used several ways
        MVC OREC+6(25),=C'*** Sem XXX BZZ9 sections'
```

(ENDDEPT is discussed below.)

In multiple level control break processing, a break at one level implies a break at all lower levels!

3. If the department has changed, then end that department and start another. We see from CHKDEPT that if the department has changed, that is, if HOLDDEPT does not equal IDEPT, then we invoke the control break routine, ENDDEPT. This new routine will (a) print the number of sections for this department (if two or more according to the specifications given above), (b) print a blank line (regardless of the number of sections for this department), and (c) reset the department-level sections counter to zero:

```

*****
*      ENDDEPT- End department                               *
*      (Process minor control break)                       *
*      Print count of sections in department                *
*      if that count is two or more.                       *
*      Regardless, skip a line.                            *
*****
ENDDEPT EQU  *
        ST  R10,SVENDDEP
        CP  #DEPT,=P'2'
        BL  ENDDEPT2
        MVC OREC(40),BLANKS  This area used several ways
        BAL R10,WRITE        Blank line before count
        MVC OREC+7(24),=C'** Dept XX BZZ9 sections'
        MVC OREC+15(2),HOLDDEPT
        MVC OREC+18(4),=X'40202120'
        ED  OREC+18(4),#DEPT
        BAL R10,WRITE
ENDDEPT2 EQU  *
        MVC OREC(40),BLANKS
        BAL R10,WRITE
        ZAP #DEPT,=P'0'      Reset counter
        L   R10,SVENDDEP
        BR  R10
```

If we wanted a count by department, whether or not that count was greater than one, then we need only remove the CP and BL instructions. (This section will not be discussed in detail as the reasoning is similar to that used in the discussion of ENDSEM in the previous program.)

4. As each record is processed, add one to the number of sections for that semester and department. This is done for each record and, as such, is included in the PROCESS routine. It is done after the check for a new semester or new department.

```
PROCESS EQU *
ST R10,SVPROC
BAL R10,CHKSEM See if new semester
BAL R10,CHKDEPT See if new department
AP #SEM,='1' Count sections by semester
AP #DEPT,='1' Count sections by department
```

5. After all records have been processed, print one last set of totals. Within the WRAPUP routine we invoke ENDSEM one last time, just as in the previous program. Note that we do not need to invoke ENDDDEPT, since ENDSEM will do that. Therefore, we will get totals for the last department (if more than 1 section was offered), as well as totals for the last semester. The routine is the same as in the previous program:

```
WRAPUP EQU *
ST R10,SVWRAP
BAL R10,ENDSEM Final control break process
CLOSE OFFER
CLOSE REPORT
```

Multiple Level Control Break - Program Solution

```
PRINT NOGEN
*****
* FILENAME: OFFER10C.MLC *
* AUTHOR : Bill Qualls *
* SYSTEM : PC/370 R4.2 *
* REMARKS : Produce list of course offerings for all *
* semesters. Includes page break logic. *
* Multiple level control break example. *
*****
START 0
REGS
BEGIN BEGIN
WTO 'OFFER10C ... Begin execution'
BAL R10,SETUP
MAIN EQU *
CLI EOFSW,C'Y'
BE EOJ
BAL R10,PROCESS
B MAIN
```

(continued)

```

EOJ      EQU      *
         BAL      R10,WRAPUP
         WTO      'OFFER10C ... Normal end of program'
         RETURN
*****
*        SETUP - Those things which happen one time only,      *
*        before any records are processed.                      *
*****
SETUP    EQU      *
         ST       R10,SVSETUP
         OI       OFFER+10,X'08'      PC/370 ONLY - Convert all
*                                               input from ASCII to EBCDIC
         OI       REPORT+10,X'08'    PC/370 ONLY - Convert all
*                                               output from EBCDIC to ASCII
         OPEN     OFFER
         OPEN     REPORT
         BAL      R10,READ
         MVC      HOLDSEM,ISEM      Major control break
         MVC      HOLDDEPT,IDEPT    Minor control break
         L        R10,SVSETUP
         BR       R10
*****
*        HDGS - Print headings.                                  *
*****
HDGS     EQU      *
         ST       R10,SVHDGS
         AP       PGS,=P'1'          Add 1 to page count
         MVC      HDPGS,=X'40202120' Edit pattern for page count
         ED       HDPGS,PGS          Move page count to heading
         PUT      REPORT,FORMFEED    PC/370 ONLY
         MVC      HDSEM,HOLDSEM
         PUT      REPORT,HD1
         PUT      REPORT,HD2
         PUT      REPORT,HD3
         PUT      REPORT,HD4
         PUT      REPORT,HD5
         ZAP      LNS,=P'0'          Reset line count to zero
         L        R10,SVHDGS
         BR       R10
*****
*        PROCESS - Those things which happen once per record.  *
*****
PROCESS  EQU      *
         ST       R10,SVPROC
         BAL      R10,CHKSEM          See if new semester
         BAL      R10,CHKDEPT        See if new department
         AP       #SEM,=P'1'         Count sections by semester
         AP       #DEPT,=P'1'        Count sections by department
         BAL      R10,CHKLNS
         BAL      R10,FORMAT
         BAL      R10,WRITE
         BAL      R10,READ
         L        R10,SVPROC
         BR       R10
*****
*        READ - Read a record.                                   *
*****
READ     EQU      *
         ST       R10,SVREAD
         GET      OFFER,IREC          Read a single offer record
         B        READX

```

(continued)

```

ATEND    EQU    *
          MVI    EOFSW,C'Y'
READX    EQU    *
          L      R10,SVREAD
          BR     R10
*****
*        CHKSEM - Check for change in semester          *
*                (major control break)                 *
*****
CHKSEM    EQU    *
          ST     R10,SVCHKSEM
          CLC    HOLDSEM,ISEM          Compare w/ current
          BE     CHKSEM                Same semester, get out
          BAL    R10,ENDSEM            Process semester break
          MVC    HOLDSEM,ISEM          Update major break field
          MVC    HOLDDEPT,IDEPT        Update minor break field
CHKSEM    EQU    *
          L      R10,SVCHKSEM
          BR     R10
*****
*        CHKDEPT - Check for change in department      *
*                (minor control break)                 *
*****
CHKDEPT    EQU    *
          ST     R10,SVCHKDEP
          CLC    HOLDDEPT,IDEPT        Compare w/ current
          BE     CHKDEPTX              Same department, get out
          BAL    R10,ENDDEPT            Process department break
          MVC    HOLDDEPT,IDEPT        Update control break field
CHKDEPTX    EQU    *
          L      R10,SVCHKDEP
          BR     R10
*****
*        ENDSEM - End semester                          *
*                (Process major control break)          *
*                Show count of sections for this semester. *
*                Force next semester to another page.    *
*****
ENDSEM    EQU    *
          ST     R10,SVENDSEM
          BAL    R10,ENDDEPT            Change in semester implies
*                change in department as well.
          MVC    OREC(40),BLANKS        This area used several ways
          MVC    OREC+6(25),=C'*** Sem XXX BZZ9 sections'
          MVC    OREC+14(3),HOLDSEM
          MVC    OREC+18(4),=X'40202120'
          ED     OREC+18(4),#SEM
          BAL    R10,WRITE
          MVC    OREC(40),BLANKS
          ZAP    #SEM,=P'0'             Reset counter
          ZAP    LNS,MAXLNS             Force next sem. to new page
          L      R10,SVENDSEM
          BR     R10
*****
*        ENDDEPT- End department                       *
*                (Process minor control break)          *
*                Print count of sections in department   *
*                if that count is two or more.          *
*                Regardless, skip a line.               *
*****

```

(continued)

```

ENDDEPT EQU *
          ST R10,SVENDDEP
          CP #DEPT,=P'2'
          BL ENDDEPT2
          MVC OREC(40),BLANKS This area used several ways
          BAL R10,WRITE Blank line before count
          MVC OREC+7(24),=C'** Dept XX BZZ9 sections'
          MVC OREC+15(2),HOLDDEPT
          MVC OREC+18(4),=X'40202120'
          ED OREC+18(4),#DEPT
          BAL R10,WRITE
ENDDEPT2 EQU *
          MVC OREC(40),BLANKS
          BAL R10,WRITE
          ZAP #DEPT,=P'0' Reset counter
          L R10,SVENDDEP
          BR R10

*****
* CHKLNLS - Check lines printed. Full page? *
*****
CHKLNLS EQU *
          ST R10,SVCHKLNLS
          CP LNS,MAXLNS
          BL CHKLNLSX
          BAL R10,HDGS
CHKLNLSX EQU *
          L R10,SVCHKLNLS
          BR R10

*****
* FORMAT - Format a single detail line. *
*****
FORMAT EQU *
          ST R10,SVFORM
          MVC OREC(40),BLANKS
          MVC OCID,ICID Course ID
          MVC OSECT,ISECT Section number
          MVC OTID,ITID Teacher ID
          MVC OCRLF,WCRLF PC/370 Only
          L R10,SVFORM
          BR R10

*****
* WRITE - Write a single detail line. *
*****
WRITE EQU *
          ST R10,SVWRITE
          PUT REPORT,OREC Write report line
          AP LNS,=P'1'
          L R10,SVWRITE
          BR R10

*****
* WRAPUP - Those things which happen one time only, *
* after all records have been processed. *
*****
WRAPUP EQU *
          ST R10,SVWRAP
          BAL R10,ENDSEM Final control break process
          CLOSE OFFER
          CLOSE REPORT
WTO 'OFFER10C ... Course list on REPORT.TXT'

```

(continued)

```

L      R10,SVWRAP
BR     R10
*****
*      Literals, if any, will go here
*****
      LTORG
*****
*      File definitions
*****
OFFER  DCB   LRECL=18,RECFM=F,MACRF=G,EODAD=ATEND,
          DDNAME='OFFER.DAT'
REPORT DCB   LRECL=42,RECFM=F,MACRF=P,
          DDNAME='REPORT.TXT'
*****
*      RETURN ADDRESSES
*****
SVSETUP DC   F'0'          SETUP
SVHDGS  DC   F'0'          HDGS
SVPROC  DC   F'0'          PROCESS
SVREAD  DC   F'0'          READ
SVFORM  DC   F'0'          FORMAT
SVWRITE DC   F'0'          WRITE
SVWRAP  DC   F'0'          WRAPUP
SVCHKLNS DC F'0'          CHKLNS
SVCHKSEM DC F'0'          CHKSEM
SVCHKDEP DC F'0'          CHKDEPT
SVENDSEM DC F'0'          ENDSEM
SVENDDEP DC F'0'          ENDEPT
*****
*      Miscellaneous field definitions
*****
WCRLF   DC   X'0D25'       PC/370 ONLY - EBCDIC CR/LF
EOFSW   DC   CL1'N'        End of file? (Y/N)
PGS     DC   PL2'0'        Nbr of pages printed.
LNS     DC   PL2'10'       Lines printed on this page.
MAXLNS  DC   PL2'10'       Max nbr lines per page.
*
BLANKS  DC   CL40' '       My line counts exclude hdgs.
HOLDSEM DC   CL3' '        Hold semester
HOLDDEPT DC CL2' '        Hold department
#SEM    DC   PL2'0'        Sections in a semester
#DEPT   DC   PL2'0'        Sections in a semester/dept
*****
*      Input record definition
*****
IREC    DS   0CL18         1-18 Offer record
ISEM    DS   CL3           1- 3 Semester
ICID    DS   0CL5          4- 8 Course ID
IDEPT   DS   CL2           4- 5 Department
        DS   CL3           6- 8 Course number
ISECT   DS   CL1           9- 9 Section number
ITID    DS   CL3          10-12 Teacher ID
IROOM   DS   CL4          13-16 Room number
IOCRLF  DS   CL2          17-18 PC/370 only - CR/LF
*****
*      Output (line) definition
*****
OREC    DS   0CL42         1-42
        DC   CL10' '       1-10
OCID    DS   CL5           11-15 Course ID

```

(continued)

```

OSECT   DC    CL6' '    16-21
        DS    CL1      22-22   Section number
        DC    CL8' '    23-30
OTID    DS    CL3      31-33   Teacher ID
        DC    CL7' '    34-40
OCRLF   DS    CL2      41-42   PC/370 only - CR/LF
*****
*      Headings definitions      *
*****
*
*  ----+----1----+----2----+----3----+----4
*          COURSE OFFERINGS      PageBZZ9
*          Semester XXX
*
*          Course      Section      Teachers
*          -----      -
*          XXXXX      X      XXX
*
*          XXXXX      X      XXX
*          XXXXX      X      XXX
*
*
*          Dept XX BZZ9 sections
*
*          Sem XXX BZZ9 sections
*
FORMFEED DS    0CL42      PC/370 only
*         DC    X'0C'      EBCDIC formfeed
*         DC    CL39' '
*         DC    40C' '      For testing...
*         DC    X'0D25'      EBCDIC CR/LF
HD1      DS    0CL42
        DC    CL36'      COURSE OFFERINGS      Page'
HDFGS    DC    CL4'BZZ9'
        DC    XL2'0D25'
HD2      DS    0CL42
        DC    CL21'      Semester '
HDSEM    DS    CL3
        DC    CL16' '
        DC    XL2'0D25'
HD3      DS    0CL42
        DC    CL40' '
        DC    XL2'0D25'
HD4      DS    0CL42
        DC    CL40'      Course      Section      Teacher      '
        DC    XL2'0D25'
HD5      DS    0CL42
        DC    CL40'      -----      -----      -----      '
        DC    XL2'0D25'
END      BEGIN

```

Rolling Totals

There is more than one way to keep the count of sections by department *and* by semester. In our solution we added one to each of these two counters in the `PROCESS` routine. It is also common to increment the minor counter only (the minor counter is, in this case, the count of sections by department, `#DEPT`). Then, when a minor control break is detected, add the minor count to the major count, before resetting the minor count to zero. This is sometimes referred to as "rolling" the totals. The implementation of this "rolling" logic would be as follows:

```

PROCESS EQU *
      ST R10,SVPROC
      BAL R10,CHKSEM
      BAL R10,CHKDEPT
      AP #DEPT,='1'

```

Have removed AP #SEM,='1'

and...

```

ENDDEPT EQU *
      .
      .
      BAL R10,WRITE
      AP #SEM,#DEPT Roll counts
      ZAP #DEPT,='0' Reset counter
      L R10,SVENDDEP
      BR R10

```

This technique is actually more efficient than the previous method. There are a total of two APs, so the program size is this same. But in the old method, #SEM is incremented (by one) once for each record. In this method, #SEM is incremented (by the amount in #DEPT) *once for each department*. The technique to be used is a matter of personal preference, but you should certainly be able to recognize and understand both.

Summary Report - Programming Problem

In our final version of the program we wish to produce a *summary* report only; that is, rather than list each course and section offered during a given semester, we will show each department and the number of sections offered by that department. Our new report will appear as follows:

COURSE OFFERINGS	Page	1
Semester W92		
Dept	Sections	
-----	-----	
EG	1	
MA	1	
PE	1	
-----	-----	
Total	3	
<hr style="border: 0.5px solid black;"/>		
COURSE OFFERINGS	Page	2
Semester F92		
Dept	Sections	
-----	-----	
AC	1	
BU	1	
EG	2	
MA	2	
PE	1	
-----	-----	
Total	7	

(continued)

COURSE OFFERINGS	Page
Semester W93	3
Dept	Sections
EG	1
MA	2
PE	1
Total	4

Summary Report - Logic

Though this report differs considerably from the previous report, the changes necessary to produce this report are really quite trivial. First, there is no record level reporting, so the branch-and-link to the `FORMAT` and `WRITE` routines have been removed from the `PROCESS` routine. In fact, the `FORMAT` routine has been removed from the program entirely. Second, the format of the counts has been changed in the `ENDSEM` and `ENDDEPT` routines. All other changes are for formatting purposes only.

Summary Report - Program Solution

```

PRINT NOGEN
*****
*      FILENAME:  OFFER10D.MLC      *
*      AUTHOR   :  Bill Qualls     *
*      SYSTEM   :  PC/370 R4.2     *
*      REMARKS  :  Produce list of course offerings for all *
*                  semesters. Includes page break logic.  *
*                  Multiple level control break.          *
*      Minor break (dept) is summary only.                *
*****
      START 0
      REGS
BEGIN  BEGIN
      WTO   'OFFER10D ... Begin execution'
      BAL  R10,SETUP
MAIN   EQU  *
      CLI  EOFSW,C'Y'
      BE   EOJ
      BAL  R10,PROCESS
      B    MAIN
EOJ    EQU  *
      BAL  R10,WRAPUP
      WTO   'OFFER10D ... Normal end of program'
      RETURN
*****
*      SETUP - Those things which happen one time only,   *
*                  before any records are processed.      *
*****
SETUP  EQU  *
      ST   R10,SVSETUP
      OI  OFFER+10,X'08'      PC/370 ONLY - Convert all
*                                  input from ASCII to EBCDIC

```

(continued)

```

*          OI    REPORT+10,X'08'    PC/370 ONLY - Convert all
*                                     output from EBCDIC to ASCII
      OPEN  OFFER
      OPEN  REPORT
      BAL   R10,READ
      MVC   HOLDSEM,ISEM            Major control break
      MVC   HOLDDEPT,IDEPT         Minor control break
      L     R10,SVSETUP
      BR    R10
*****
*          HDGS - Print headings.
*****
HDGS    EQU    *
      ST     R10,SVHDGS
      AP     PGS,=P'1'              Add 1 to page count
      MVC   HDPGS,=X'40202120'    Edit pattern for page count
      ED     HDPGS,PGS             Move page count to heading
      PUT   REPORT,FORMFEED       PC/370 ONLY
      MVC   HDSEM,HOLDSEM
      PUT   REPORT,HD1
      PUT   REPORT,HD2
      PUT   REPORT,HD3
      PUT   REPORT,HD4
      PUT   REPORT,HD5
      ZAP   LNS,=P'0'              Reset line count to zero
      L     R10,SVHDGS
      BR    R10
*****
*          PROCESS - Those things which happen once per record.
*****
PROCESS EQU    *
      ST     R10,SVPROC
      BAL   R10,CHKSEM             See if new semester
      BAL   R10,CHKDEPT           See if new department
      AP     #SEM,=P'1'           Count by semester
      AP     #DEPT,=P'1'          Count by semester/dept
      BAL   R10,READ              No detail-level reporting
      L     R10,SVPROC            Removed BAL to FORMAT and WRITE
      BR    R10
*****
*          READ - Read a record.
*****
READ    EQU    *
      ST     R10,SVREAD
      GET   OFFER,IREC           Read a single offer record
      B     READX
ATEND   EQU    *
      MVI   EOFSW,C'Y'
READX   EQU    *
      L     R10,SVREAD
      BR    R10
*****
*          CHKSEM - Check for change in semester
*          (major control break)
*****
CHKSEM  EQU    *
      ST     R10,SVCHKSEM
      CLC   HOLDSEM,ISEM         Compare w/ current
      BE    CHKSEM               Same semester, get out
      BAL   R10,ENDSEM           Process semester break

```

(continued)

```

                MVC   HOLDSEM,ISEM           Update major break field
                MVC   HOLDDEPT,IDEPT        Update minor break field
CHKSEM   EQU   *
                L     R10,SVCHKSEM
                BR    R10
*****
*           CHKDEPT - Check for change in department           *
*                   (minor control break)                     *
*****
CHKDEPT   EQU   *
                ST    R10,SVCHKDEP
                CLC   HOLDDEPT,IDEPT        Compare w/ current
                BE    CHKDEPTX              Same department, get out
                BAL   R10,ENDDEPT           Process department break
                MVC   HOLDDEPT,IDEPT        Update control break field
CHKDEPTX  EQU   *
                L     R10,SVCHKDEP
                BR    R10
*****
*           ENDSEM - End semester                             *
*                   (Process major control break)             *
*                   Show count of sections for this semester. *
*                   Force next semester to another page.     *
*****
ENDSEM    EQU   *
                ST    R10,SVENDSEM
                BAL   R10,ENDDEPT           Change in semester implies
*                                               change in department as well.

                PUT   REPORT,HD5
                AP    LNS,=P'1'
                MVC   OREC(40),BLANKS      Reuse this line
                MVC   ODEPT(5),=CL5'Total'
                MVC   O#DEPT,=X'40202120'
                ED    O#DEPT,#SEM
                BAL   R10,WRITE
                MVC   OREC(40),BLANKS
                ZAP   #SEM,=P'0'           Reset semester record count
                ZAP   LNS,MAXLNS           Force next sem. to new page
                L     R10,SVENDSEM
                BR    R10
*****
*           ENDDEPT- End department                             *
*                   (Process minor control break)             *
*                   Print a count of courses (sections) in    *
*                   this department.                           *
*****
ENDDEPT   EQU   *
                ST    R10,SVENDDEP
                BAL   R10,CHKLNS
                MVC   OREC(40),BLANKS
                MVC   ODEPT,HOLDDEPT      Department
                MVC   O#DEPT,=X'40202120'
                ED    O#DEPT,#DEPT      How many this department?
                MVC   OCRLF,WCRLF      PC/370 only
                BAL   R10,WRITE
                ZAP   #DEPT,=P'0'           Reset dept record count
                L     R10,SVENDDEP
                BR    R10
*****
*           CHKLNS - Check lines printed. Full page?         *
*****

```

(continued)

```

CHKLNS  EQU  *
        ST   R10,SVCHKLNS
        CP   LNS,MAXLNS
        BL   CHKLNSX
        BAL  R10,HDGS
CHKLNSX EQU  *
        L    R10,SVCHKLNS
        BR   R10
*****
*       WRITE - Write a single detail line.          *
*****
WRITE   EQU  *
        ST   R10,SVWRITE
        PUT  REPORT,OREC          Write report line
        AP   LNS,=P'1'
        L    R10,SVWRITE
        BR   R10
*****
*       WRAPUP - Those things which happen one time only, *
*               after all records have been processed.    *
*****
WRAPUP  EQU  *
        ST   R10,SVWRAP
        BAL  R10,ENDSEM          Final control break process
        CLOSE OFFER
        CLOSE REPORT
*****
WTO     'OFFER10D ... Course list on REPORT.TXT'
        L    R10,SVWRAP
        BR   R10
*****
*       Literals, if any, will go here                *
*****
        LTORG
*****
*       File definitions                               *
*****
OFFER   DCB  LRECL=18,RECFM=F,MACRF=G,EODAD=ATEND,
          DDNAME='OFFER.DAT'
REPORT  DCB  LRECL=42,RECFM=F,MACRF=P,
          DDNAME='REPORT.TXT'
*****
*       RETURN ADDRESSES                             *
*****
SVSETUP DC  F'0'          SETUP
SVHDGS  DC  F'0'          HDGS
SVPROC  DC  F'0'          PROCESS
SVREAD  DC  F'0'          READ
SVFORM  DC  F'0'          FORMAT
SVWRITE DC  F'0'          WRITE
SVWRAP  DC  F'0'          WRAPUP
SVCHKLNS DC F'0'          CHKLNS
SVCHKSEM DC F'0'          CHKSEM
SVCHKDEP DC F'0'          CHKDEPT
SVENDSEM DC F'0'          ENDSEM
SVENDDEP DC F'0'          ENDDDEPT
*****
*       Miscellaneous field definitions              *
*****
WCRLF   DC  X'0D25'       PC/370 ONLY - EBCDIC CR/LF
EOF5W   DC  CL1'N'        End of file? (Y/N)
PGS     DC  PL2'0'        Nbr of pages printed.

```

(continued)

```

LNS      DC      PL2'10'          Lines printed on this page.
MAXLNS   DC      PL2'10'          Max nbr lines per page.
*                                               My line counts exclude hdgs.
BLANKS   DC      CL40' '
HOLDSEM  DC      CL3' '          Hold semester
HOLDDEPT DC      CL2' '          Hold department
#SEM     DC      PL2'0'          Sections in a semester
#DEPT    DC      PL2'0'          Sections in a semester/dept
*****
*          Input record definition          *
*****
IREC     DS      0CL18      1-18   Offer record
ISEM     DS      CL3        1- 3   Semester
ICID     DS      0CL5      4- 8   Course ID
IDEPT    DS      CL2        4- 5   Department
         DS      CL3        6- 8   Course number
ISECT    DS      CL1        9- 9   Section number
ITID     DS      CL3       10-12   Teacher ID
IROOM    DS      CL4       13-16   Room number
IOCRLF   DS      CL2       17-18   PC/370 only - CR/LF
*****
*          Output (line) definition        *
*****
OREC     DS      0CL42      1-42
         DC      CL10' '      1-10
ODEPT    DS      CL2        11-12  Department
         DC      CL7' '      13-19
O#DEPT   DS      CL4        20-23  Number of sections
         DC      CL17' '     24-40
OCRLF    DS      CL2       41-42   PC/370 only - CR/LF
*****
*          Headings definitions            *
*****
*
*  ----+----1----+----2----+----3----+----4
*          COURSE OFFERINGS          PageBZZ9
*          Semester XXX
*
*
*          Dept          Sections
*          -----
*          XX           BZZ9
*          XX           BZZ9
*          -----
*          Total        BZZ9
*
FORMFEED DS      0CL42          PC/370 only
*         DC      X'0C'          EBCDIC formfeed
*         DC      CL39' '
*         DC      40C' '          For testing...
*         DC      X'0D25'          EBCDIC CR/LF
HD1      DS      0CL42
         DC      CL36'          COURSE OFFERINGS      Page'
HDFGS    DC      CL4'BZZ9'
         DC      XL2'0D25'
HD2      DS      0CL42
         DC      CL21'          Semester '
HDSEM    DS      CL3
         DC      CL16' '
         DC      XL2'0D25'

```

(continued)

```
HD3      DS      0CL42
         DC      CL40' '
         DC      XL2'0D25'
HD4      DS      0CL42
         DC      CL40'          Dept      Sections          '
         DC      XL2'0D25'
HD5      DS      0CL42
         DC      CL40'          -----  -----          '
         DC      XL2'0D25'
         END     BEGIN
```

Exercises

1. True or false.
 - T F a. In control break processing, the control break field(s) are initialized in `SETUP` prior to the priming read.
 - T F b. When multiple level control breaks are used, the control break fields are evaluated minor before major.
 - T F c. When multiple level control breaks are used, a major break implies a minor break.
 - T F d. Rather than increment counters at all levels within `PROCESS`, the major totals can be "rolled" into the minor totals.
 - T F e. When a control break is detected, counters or totals are rolled before they are zeroed.
 - T F f. Multiple level control break processing requires two hold fields for each level.
 - T F g. A simple technique for forcing a page break in control break processing is to set the page counter to the maximum allowed.
 - T F h. Adding control break logic to a program will necessitate a change to the mainline.
 - T F i. Adding control break logic to a program will necessitate a change to the `WRAPUP` routine.
 - T F j. Adding control break logic to a program will necessitate a change to the report `DCB`.
 - T F k. Changes in the control break fields are checked prior to formatting the report detail line for the current record.
 - T F l. Control break logic can be used in summary reports as well as in detail reports.
 - T F m. A separate line counter is needed for each level of control break processing.

2. (Refer to the Small Town Hardware Store database in More Datasets.) Produce a list of the tools found in each kit. Allow 10 lines per page. Your output should appear as follows:

```

          1         2         3         4
1234567890123456789012345678901234567890
-----
SMALL TOWN HARDWARE                               Page BZZ9

  Kit ID      Tool ID
  -----      -
  XXX          XXX
              XXX

  XXX          XXX
              XXX
              XXX
  
```

Exercises

The kit ID should be printed on the first line for that kit, and on the first line for a page if that kit's listing should span a page break. Double space between kits.

3. (Refer to the Small Town Blood Bank database in More Datasets.) Use the donation file to produce a group history summary as follows:

1	2	3
1234567890	1234567890	1234567890
SMALL TOWN BLOOD BANK		
Group ID	Number of Donations	
-----	-----	
XXX	BZZ9	
XXX	BZZ9	
XXX	BZZ9	
-----	-----	
Total	BZZ9	

In order to produce this report, the file will need to be sorted by group ID. DOS' `SORT` command is crude but will suffice. Type the following at the DOS prompt:

```
sort /+4 < donation.dat > donation.srt
```

This will create a new file, `DONATION.SRT`, which is sorted beginning with the fourth column of each record. Specify `DONATION.SRT` as the `DDNAME` for the input `DCB`.

4. (Refer to the Small Town Blood Bank database in More Datasets.) Produce a donor history summary as follows:

1	2	3	4	5
1234567890	1234567890	1234567890	1234567890	1234567890
SMALL TOWN BLOOD BANK			Page BZZ9	
Donor ID	First Donation	Last Donation	Number of Donations	
-----	-----	-----	-----	
XXX	mm/dd/yy	mm/dd/yy	BZZ9	
XXX	mm/dd/yy	mm/dd/yy	BZZ9	
XXX	mm/dd/yy	mm/dd/yy	BZZ9	
-----	-----	-----	-----	
Total			BZZ9	

In order to produce this report, the file will need to be sorted by donor ID. DOS' `SORT` command is crude but will suffice. Type the following at the DOS prompt:

```
sort /+1 < donation.dat > donation.srt
```

This will create a new file, `DONATION.SRT`, which is sorted beginning with the first column of each record. Specify `DONATION.SRT` as the `DDNAME` for the input `DCB`.

Exercises

(Hint: reformat the dates from `mmddy` to `ymmdd`. The minimum date for a particular donor will be the first donation date, and the maximum date will be the last donation date.)

5. (Refer to the Small Town Payroll database in More Datasets.) Use the `HISTORY` file to produce a payroll register for all pay periods. The report should appear as follows, with *one pay period per page*:

```

      1         2         3         4         5
12345678901234567890123456789012345678901234567890
SMALL TOWN PAYROLL
Payroll Register for PPED mm/dd/yy

Employee      Hours      Gross
-----
   XXX      BZZ9.99      BZZ,ZZ9.99
   XXX      BZZ9.99      BZZ,ZZ9.99
   XXX      BZZ9.99      BZZ,ZZ9.99
-----
TOTAL          BZZ9.99      BZZ,ZZ9.99
```

There were BZZ9 checks printed for PPED mm/dd/yy.